





Programming Issues;

'==========================================================



Please don't take offence by anything said herein, nothing mentioned here is meant as a criticism of anyone. I hope this article will be seen as a positive 'push' to encourage every programmer to continually improve their craft.



No doubt some of the things said here will already be part of your programming code of practice, if so please bare with me. Reminders can be helpful too.



Code written in GFA can be as good as any other language (or better) provided one takes pride in and spends time on one's code.  



There are also some hints and tips.



J.M.F



'===================================================================





 

1. Always use meaningful variable names.





It's OK in a FOR loop but most variables should be named appropriately.



    nSystemFontHeight = @GetSystemFontHeight(hWnd)







2. Use white space horizontally. 



Space out parameters horizontally so that you get both clarity and in some cases tabulation. It is easier to read and easier for others to follow.



 PROCEDURE MenuProc(hWnd, hMenu)



   ~EnableMenuItem(hMenu, IDM_SAVE,    MF_GRAYED)

   ~EnableMenuItem(hMenu, IDM_PALETTE, MF_GRAYED)

   ~EnableMenuItem(hMenu, IDM_CUT,     MF_GRAYED)

   ~EnableMenuItem(hMenu, IDM_COPY,    MF_GRAYED)

   ~EnableMenuItem(hMenu, IDM_PASTE,   MF_GRAYED)



   {lprgbq + n * 4}.rgbRed      = iR





3. Use white space vertically.



Use white space vertically to separate different sections of code. 



      SELECT



      CASE IDM_PRINT

        @PrintJob(hWnd)



      CASE IDM_ABOUT

        _CB (8)   = AboutModal(wwwl)

        hDlgModal = DialogBox(hInstance, "about", hWnd, _CB(8))



      CASE IDM_EXIT

        ~PostMessage(hWnd, WM_CLOSE, 0, 0)



      ENDSELECT







4.  It helps to use a form of Hungarian Notation. 



Hungarian Notation means that you use prefix's to variables that carry meaning; an example would be to use lp for a long pointer, h for a handle etc. Capitalise the start of a word within a variable name, i.e. bIsWindowActive



    lpMem = MALLOC(1024)



    hBrush = CreateSolidBrush({lParam}.itemData)



    bIsWindowActive = FALSE       b is for Boolean



    fMagFact        = 2.17        f is for float



    etc.



You may even prefer it to postfix's.



Use the GFA define commands;

 

DEFDBL, DEFINT, DEFWRD, DEFBYT, DEFSTR, DEFBIT







5.  Use Uppercase characters for constants with notation.



    DEFBYT "ID"                     //  8 bit, byte



    IDM_SAVE  = 80               ID is identifier,  M is menu



    IDB_SAVE  = 110                "   "         ,  B is button







6.  Use Local variables where possible.



Design your code so that most variables are local to a sub-routine. Only use global variables when necessary.



This can also include TYPE's or Structures. DIM the TYPE at the start at a sub-routine;



    LOCAL hdcMem, hOldObject, nOffSet

    BITMAPINFOHEADER: bi.



 and erase the TYPE before leaving the sub-routine



    ERASE bi.







7.  Take pride in your code.



For G-d's sake (no, I'm not Jewish) write more REM's.



Especially if others will see your code, and I hope they will, its necessary to explain what the code is doing. It may be abundantly clear to you, you wrote it, but others will not see things as clearly.



The number of lines that adding REM's will produce doesn't matter a jot!



Your editor won't complain, honest!





 

8.  Declare all global variables at the start of your programme.



This is most often done in a PROCEDURE Initialise().  REM the variables;



    nMAXPALETTE  = 256     // maximum supported palette entries



    hBitMap      = 0       // used in DrawBitMap()







9.  Use TYPE element offsets instead of integer calculation.



For example when filling the red, green and blue bytes for palettes one could use;



    FOR n = 0 TO 255

      other programme lines



      BYTE{lprgbq + 4 * n + 0} = iB

      BYTE{lprgbq + 4 * n + 1} = iG

      BYTE{lprgbq + 4 * n + 2} = iR

      BYTE{lprgbq + 4 * n + 3} = 0

    NEXT n



But the better solution is to use the element offsets of the TYPE RGBQUAD:



   FOR n = 0 TO 255

      other programme lines



     {lprgbq + n * 4}.rgbRed      = iR

     {lprgbq + n * 4}.rgbGreen    = iG

     {lprgbq + n * 4}.rgbBlue     = iB

     {lprgbq + n * 4}.rgbReserved = 0

   NEXT n



Not only is it more readable but if MS change a structure, which they do sometimes, your code will be easier to change as well.







10.  Shorter Routines.



Try not to have very long PROCEDURES or FUNCTIONS. Frequently the SELECT ENDSELECT parts of a MainWndProc procedure can become large. One way this can be avoided is to move the SELECT ENDSELECT after WM_COMMAND or any other WM_ message to another sub-routine; there could be many messages to respond to.



A better solution, something like this;



  WM_COMMAND

    @DoCommands(hWnd, wMessage, wParam, lParam)



  WM_CLOSE







11. Release 'Objects' after use.



Windows has finite resources, although with the advent of Windows 95 things have improved, the same care should be exercised when creating and deleting 'objects' as with Win 3.1



Always cleanup before you application terminates; it used to be called 'Exiting Gracefully'.



PROCEDURE Cleanup()

  IF hMyObject THEN ~DeleteObject(hMyObject)

    other programme lines

RETURN



Normally one would create 'objects' when the application is initialising but sometimes you need to create many 'objects' as the programme is running. Here is a routine that I have found useful.



hBrush = @CreateBrush(hBrush, RGB(255, 0, 0))



FUNCTION CreateBrush(hBrush, lCol)

  IF hBrush THEN ~DeleteObject(hBrush)

  RETURN CreateSolidBrush(lCol)

ENDFUNC



The existing BRUSH (object) is always deleted before the new one is created.







12. Using Resources.



Resources are a convenient way to manage your Menu's, Dialog's, String Table's, Icon's, Bitmap's, Cursor's and Keyboard Accelerators.



If you have a resource compiler GFA will happily link the resource at compile time. 



$LNK RES "d:\project\myres.res"



You can then load or manage the resources simply by using the appropriate Windows API call.



AccessResource    Opens an executable file and locates a resource 

AllocResource     Allocates memory for a resource 

FindResource      Locates a resource in a resource file 

FreeResource      Frees a loaded resource 

LoadAccelerators  Loads an accelerator table 

LoadBitmap        Loads a bitmap resource 

LoadCursor        Loads a cursor resource 

LoadIcon          Loads an icon resource 

LoadMenu          Loads a menu resource 

LoadResource      Loads the specified resource in global memory 

LoadString        Loads a string resource 

LockResource      Locks a resource in memory



Example;



hMenu = LoadMenu(hInstance, "MENU_1")

~SetMenu(hWnd, hMenu)



This will only be possible for the compiled programme. However, the same thing can be achieved for an Interpreted programme by substituting a DLL for the RES file.



Find out if the programme is running interpreted or compiled;



myDllPath$ = "e:\myproj\myproj.dll"

GFAPath$   = "D:\GFAWIN\GFAWIN37.EXE"



bComp = @CompOrNot()                ' Compiled or Not



FUNCTION  CompOrNot()

  LOCAL Name$

  ' -------------------------------------------------------------------

  ' Either the res or dll will be used to load resources dependant on if

  ' compiled or not.

  ' -------------------------------------------------------------------

  Name$ = SPACE$(244)



  ~GetModuleFileName(_INSTANCE, V:Name$, 240)



  Name$ = TRIM$(Name$)



  GFAPath$ = UPPER$(GFAPath$)



  '--------------------------

  ' Interpreted

  '--------------------------

  IF LEFT$(Name$, LEN(Name$) - 1) = GFAPath$

    hInstRes = LoadLibrary(myDllPath$)

    RETURN FALSE



    '--------------------------

    ' Compiled

    '--------------------------

  ELSE

    hInstRes = _INSTANCE

    RETURN TRUE

  ENDIF



ENDFUNC



The global variable  hInstRes  will always be correct whether you run interpreted or compiled so the limitation of only running compiled when using resources does not apply. No more excuses!!!



You would then use the variable hInstRes to load resources;



hMenu = LoadMenu(hInstRes, "MENU_1")



And in your PROCEDURE Cleanup()



IF !bComp THEN ~FreeLibrary(hInstRes)







13. Creating a Resource Only DLL.



Create the DLL either by loading your res into the resource compiler and then overwriting an existing DLL, or by the method below;



Compile these lines only;



$LIBRARY

$DESCR GFA-Compiled Resource DLL

$LNK EXE d:\myproj.dll

$LNK RES d:\myproj.res



This will link the file d:\myproj.res and create a resource dll called d:\myproj.dll



You can then use it to load resources when running Interpreted.





Use CmdShow.



It is proper to use the start-up Window setting when creating your main window. The user sets this with ‘Properties’.



Although GFA does not support this feature you can find the value of CmdShow with WORD{_PSP + $5E}. It will be one of these values;




SW_SHOWMINIMIZED

SW_SHOWMAXIMIZED

SW_SHOWNORMAL




You can also find the long pointer to the command line string using;

 _PSP + $81



So, putting all this together; pass these values to the functions that will use them.



@WinMain(_PSP + $81, WORD{_PSP + $5E})



PROCEDURE WinMain(lpszCmdLine, nCmdShow)

  Cmd$ = CHAR{lpszCmdLine}



  IF !@InitApplication(_INSTANCE) THEN @Cleanup(1)

  IF !@CreateMainWindow(nCmdShow) THEN @Cleanup(2)



After the Window is created use  ~ShowWindow(hWndMain, nCmdShow)







Not Enough Callbacks.



Although unlikely for most purposes you may need more then 16 callbacks. 



There is a way this can be done from within GFA.



This information was gleaned from Mr. Eberhard Funck



For the PROCEDURE that needs a callback using this method, you put the $Export directive in that PROC like so;



PROCEDURE MainWndProc(hWnd, wMessage, wParam, lParam)

  $Export MainWndProc



And when creating the Window Class, set the long pointer to Window Proc with this;



wndClass.lpFNWndProc  = GetProcAddress(NULL, "MainWndProc")



This will only work compiled but in the rare circumstance that you run out of callbacks it will be a boon.



Another example for Dialogs will suffice;



If the programme is compiled (bComp = TRUE), use



GetProcAddress(NULL, "AboutModal") to set the pointer for the Dialog Procedure address.



When Interpreted, use the normal GFA _CB(x)



Don’t forget the $Export directive in the ABOUT procedure.



PROCEDURE AboutModal (hWdlg, wMessage, wParam, lParam)

  $Export AboutModal



CASE IDM_ABOUT

  IF bComp

    hDlgModal = DialogBox(hInstRes, "about", hWnd, GetProcAddress(NULL,

                                                     "AboutModal"))

  ELSE

    _CB (8) = AboutModal(wwwl) 

    hDlgModal = DialogBox (hInstRes, "about", hWnd, _CB(8))



  ENDIF



There is an example listing to illustrate the various things discussed in this document.



When selecting a menu item from the example listing, API LoadString loads the string from the resource; the Window Icon and the DailogBox are also loaded from the resource.



Makes sure you set the paths correctly



$LNK EXE  e:\awb\Example.exe

$LNK RES "e:\awb\Example.res"



and



zDllPath = "e:\awb\example.dll"         ' Resource DLL's full path

zGFAPath = "D:\GFAWIN\GFAWIN37.EXE"     ' The Interpreters full path



J.M.F Mar. 1999











